Problem:
Score : 500 points

Problem Statement

Takahashi has a string S of length N consisting of digits from 0 through 9.

He loves the prime number P. He wants to know how many non-empty (contiguous) substrings of S - there are N \times (N + 1) / 2 of them - are divisible by P when regarded as integers written in base ten.

Here substrings starting with a 0 also count, and substrings originated from different positions in S are distinguished, even if they are equal as strings or integers.

Compute this count to help Takahashi.

Constraints

1 \leq N \leq 2 \times 10^5

S consists of digits.

|S| = N

2 \leq P \leq 10000

P is a prime number.

Input

Input is given from Standard Input in the following format:

N P
S

Output

Print the number of non-empty (contiguous) substrings of S that are divisible by P when regarded as an integer written in base ten.

Sample Input 1

4 3
3543

Sample Output 1

6

Here S = 3543. There are ten non-empty (contiguous) substrings of S:

3: divisible by 3.

35: not divisible by 3.

354: divisible by 3.

3543: divisible by 3.

5: not divisible by 3.

54: divisible by 3.

543: divisible by 3.

4: not divisible by 3.

43: not divisible by 3.

3: divisible by 3.

Six of these are divisible by 3, so print 6.

Sample Input 2

4 2
2020

Sample Output 2

10

Here S = 2020. There are ten non-empty (contiguous) substrings of S, all of which are divisible by 2, so print 10.

Note that substrings beginning with a 0 also count.

Sample Input 3

20 11
33883322005544116655

Sample Output 3

68

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program main
  implicit none
  integer(8) :: n, m, i, dig, res
  integer(8), allocatable :: d(:), cnt(:)
  character(200000) :: s
  read *, n, m
  read *, s
  allocate (d(n), cnt(0:m - 1))

  if (m == 2 .or. m == 5) then
    res = 0
    do i = 1, n
      read (s(i:i), *) dig
      if (modulo(dig, m) == 0) res = res + i
    end do
  else
    read (s(n:n), *) dig
    cnt = 0
    cnt(0) = 1
    d(n) = modulo(dig, m)
    cnt(d(n)) = cnt(d(n)) + 1
    do i = n - 1, 1, -1
      read (s(i:i), *) dig
      d(i) = d(i + 1) + pow(10_8, n - i) * dig
      d(i) = modulo(d(i), m)
      cnt(d(i)) = cnt(d(i)) + 1
    end do

    res = 0
    do i = 0, m - 1
      res = res + cnt(i) * (cnt(i) - 1) / 2
    end do
  end if
  print "(i0)", res
  contains

  recursive function pow(n, p) result(res)
    integer(8), intent(in) :: n, p
    integer(8) :: res
    if (p == 0) then
      res = 1
    else if (modulo(p, 2_8) == 0) then
      res = modulo(pow(n, p / 2) ** 2, m)
    else
      res = modulo(pow(n, p - 1) * n, m)
    end if
  end function pow
end program main